

-- BcdHeap.Mesa Edited by Johnsson on February 26, 1978 12:20 PM

DIRECTORY

AltoDefs: FROM "altodefs",
 BcdHeapDefs: FROM "BcdHeapDefs",
 FSPDefs: FROM "fspdefs",
 StringDefs: FROM "stringdefs",
 SystemDefs: FROM "systemdefs";

DEFINITIONS FROM FSPDefs;

BcdHeap: PROGRAM

IMPORTS FSPDefs, StringDefs, SystemDefs EXPORTS BcdHeapDefs = PUBLIC
 BEGIN

Heap: PRIVATE ZonePointer ← NIL;

GetSpace: PROCEDURE [nwords: CARDINAL] RETURNS [p: POINTER] =
 BEGIN OPEN SystemDefs;
 np: CARDINAL;
 p ← MakeNode[Heap, nwords
 !NoRoomInZone =>
 BEGIN
 np ← MAX[PagesForWords[nwords + ZoneOverhead + 1--NodeOverhead--], 2];
 AddToNewZone[Heap, AllocatePages[np], np*AltoDefs.PageSize, FreePages];
 RESUME
 END];
 RETURN
 END;

FreeSpace: PROCEDURE [p: POINTER] =
 BEGIN
 FreeNode[Heap, p]; RETURN
 END;

GetString: PROCEDURE [nchars: CARDINAL] RETURNS [p: STRING] =
 BEGIN
 OPEN StringDefs;
 p ← GetSpace[WordsForString[nchars]];
 p↑ ← [length:0, maxlength:nchars, text:];
 RETURN
 END;

FreeString: PROCEDURE [s: STRING] = LOOPHOLE[FreeSpace];

InitHeap: PROCEDURE =
 BEGIN
 IF Heap # NIL THEN EraseHeap[];
 Heap ←
 MakeNewZone[SystemDefs.AllocatePages[8], 8*AltoDefs.PageSize, SystemDefs.FreePages];
 RETURN
 END;

EraseHeap: PROCEDURE =
 BEGIN
 DestroyZone[Heap];
 Heap ← NIL;
 RETURN
 END;

END.